Le Mans User Optimisation Group: Shade values in a report using a format.

Posted by PHILIP MASON 30-Aug-2016

You can create a format to return a colour based on a value. That colour can then be used to colour a cell in a report based on the value in that cell. This approach is sometimes referred to as traffic lighting, especially when used with a small range of colours such as Red, Amber & Green. However you can use lots of colours. I like to use a gradually changing range of colours, which means you can visually understand data in a table quickly and easily.

SAS Code

(you can copy this and run it in Enterprise Guide)

```
%* Make a format to provide a smooth shading from one color to another based on values of a variable
   in a dataset - so that we can visually see the variations in values ;
%macro shade_tab_fmt(dset,var,fmtname) ;
   %* dset = datatset name ;
   %* var = variable to use ;
   %* fmtname = name of format to create ;
   * Work out the min, max & range of the variable we are using ;
   proc sql noprint ;
      select min(&var),max(&var),range(&var) into :shade_min,:shade_max,:shade_range from &dset ;
   quit ;
   %* Set each component of the color to produce a smooth range of colors when mixed as RGB components.
      You can change the code here to create different kinds of colour ranges. ;
   %let red=put(i/&shade_max*255,hex2.) ;
   %let green=put((&shade_max-i)/&shade_max*255,hex2.) ;
   %let blue='20' ; * put(i/&shade_max*255,hex2.) ;
 * Build a format to shade cells from one color to another ;
   data _null_ ;
      call execute("proc format ; value &fmtname") ;
      max=1;
      steps=64 ; * number of steps in color gradation ;
```

```
      step=&shade_range/steps ;
      do i=&shade_min to &shade_max by step ;
        put i= ;
        color='cx'||&red||&green||&blue ;
        from=i-step/2 ;*((i-1)/&shade_max)*max ;
        to=i+step/2 ; *(i/&shade_max)*max ;
        call execute(put(from,best.)||'-'||put(to,best.)||'='||quote(color)) ;
        end ;
      call execute('.="light gray" other="cxd0d0d0" ; run ;') ;
   run ;
%mend shade_tab_fmt ;
* Make a format that is shaded according to values ;
%shade_tab_fmt(sashelp.retail, sales, shade) ;
* Make a report using the format ;
proc tabulate data=sashelp.retail style={background=shade.} ;
   class year month ;
   var sales ;
   table sales*sum,year,month ;
run ;
* Make a report using format with Proc Report ;
data air ;
  set sashelp.air ;
   year=year(date) ;
   month=month(date) ;
run ;
%shade_tab_fmt(sashelp.air, air, airfmt) ;
proc report data=air ;
   column year air,month ;
   define year / group ;
   define month / across  ;
   define air / style(column)= [background=airfmt.] ;
run ;
```

Reports Produced

Sum of Retail sales in millions of $				
	MONTH			
	1	4	7	10
YEAR				
1980	220.00	257.00	258.00	295.00
1981	247.00	292.00	286.00	323.00
1982	284.00	307.00	318.00	343.00
1983	299.00	351.00	359.00	384.00
1984	342.00	388.00	385.00	413.00
1985	337.00	399.00	412.00	448.00
1986	419.00	472.00	490.00	541.00
1987	484.00	543.00	542.00	595.00
1988	546.00	607.00	616.00	643.00
1989	594.00	666.00	662.00	670.00
1990	606.00	674.00	705.00	749.00
1991	703.00	709.00	728.00	807.00
1992	692.00	797.00	826.00	889.00
1993	758.00	909.00	920.00	991.00
1994	876.00	998.00	.	.

	international airline travel (thousands)											
	month											
year	1	2	3	4	5	6	7	8	9	10	11	12
1949	112	118	132	129	121	135	148	148	136	119	104	118
1950	115	126	141	135	125	149	170	170	158	133	114	140
1951	145	150	178	163	172	178	199	199	184	162	146	166
1952	171	180	193	181	183	218	230	242	209	191	172	194
1953	196	196	236	235	229	243	264	272	237	211	180	201
1954	204	188	235	227	234	264	302	293	259	229	203	229
1955	242	233	267	269	270	315	364	347	312	274	237	278
1956	284	277	317	313	318	374	413	405	355	306	271	306
1957	315	301	356	348	355	422	465	467	404	347	305	336
1958	340	318	362	348	363	435	491	505	404	359	310	337
1959	360	342	406	396	420	472	548	559	463	407	362	405
1960	417	391	419	461	472	535	622	606	508	461	390	432

22 Views

NARENDRA KALAPALA

31-Aug-2016 10:47

Amazing stuff. Thanks for sharing.